# Copyright 2021 Louis Bettens
#
# Permission to use, copy, modify, and/or distribute this software for any
# purpose with or without fee is hereby granted, provided that the above
# copyright notice and this permission notice appear in all copies.
#
# THE SOFTWARE IS PROVIDED "AS IS" AND THE AUTHOR DISCLAIMS ALL WARRANTIES
# WITH REGARD TO THIS SOFTWARE INCLUDING ALL IMPLIED WARRANTIES OF
# MERCHANTABILITY AND FITNESS. IN NO EVENT SHALL THE AUTHOR BE LIABLE FOR
# ANY SPECIAL, DIRECT, INDIRECT, OR CONSEQUENTIAL DAMAGES OR ANY DAMAGES
# WHATSOEVER RESULTING FROM LOSS OF USE, DATA OR PROFITS, WHETHER IN AN
# ACTION OF CONTRACT, NEGLIGENCE OR OTHER TORTIOUS ACTION, ARISING OUT OF
# OR IN CONNECTION WITH THE USE OR PERFORMANCE OF THIS SOFTWARE.

inputs@{ nixpkgs, ... }:
{ baseModules, config, lib, pkgs, specialArgs, ... }:
let
  hostPkgs = pkgs; # give the guest the packages from the host configuration
  mkGuest = name: configuration: system: nixpkgs.lib.nixosSystem {
    inherit lib baseModules specialArgs;
    inherit system;
    modules = [
      configuration
      ../core
      {
        boot.miniguest.enable = true;
        networking.hostName = lib.mkDefault name;
        nixpkgs.pkgs = hostPkgs;
      }
    ];
  };
in
lib.mkIf (config.miniguests != { }) {
  environment.etc =
    with lib.attrsets;
    lib.flip mapAttrs' config.miniguests
      (name: { configuration, system, ... }: nameValuePair "miniguests/${name}" {
        source = (mkGuest name configuration system).config.system.build.miniguest;
      });
}
